



# Programação Estruturada

Professor Adjenor Cristiano  
Queiroz  
FAPAM - 2023

# Aula 4



HORA DA  
CORREÇÃO!



1 – Desenvolva um software que receba uma idade do usuário e imprima na tela se ele é maior de idade ou menor de idade.

Área dos algoritmos ( Edição do código fonte ) -> Nome do arquivo: [IDADE.ALG]

```
1 Algoritmo "Idade"
2 Var
3   idade : inteiro
4 Inicio
5
6   escreval("Qual sua Idade? ")
7   leia(idade)
8   se(idade>=18) entao
9     escreval("Maior de Idade ")
10  senao
11    escreval("Menor de Idade ")
12  fimse
13
14 Fimalgoritmo
15
```

2 - Desenvolva um software que receba um nome e o sexo do usuário, após imprima na tela uma saudação conforme o sexo do usuário, EX: "Bem Vindo José" ou "Bem Vinda Maria".

Área dos algoritmos ( Edição do código fonte ) -> Nome do arquivo: [IDADE.ALG]

```
1 Algoritmo "Saudação"
2 Var
3   nome, sexo : caractere
4 Inicio
5
6   escreval("Qual seu Nome? ")
7   leia(nome)
8   escreval("Qual seu Sexo? [M/F]")
9   leia(sexo)
10  se ( (sexo="F") OU (sexo="f") ) entao
11    escreval("Bem Vinda ", nome)
12  senao
13    escreval("Bem Vindo  ", nome)
14  fimse
15
16 Fimalgoritmo
17
```

3 - Doar sangue é uma atitude muito importante, entretanto é preciso atender alguns requisitos para poder doar sangue.

- Ter entre 19 e 69 anos de idade
- Pesar ao menos 50 kg
- Não ter feito nenhuma tatuagem no último ano
- Não ter ingerido álcool nas últimas 12 horas.

Desenvolva um software que pergunte ao usuário se ele atende os requisitos necessários para doar sangue. Faça as perguntas uma por vez, caso a resposta dele o impeça de doar, avise-o e encerre o sistema. Se ele atender todos os requisitos, avise-o que ele pode ser doador.

```
1 Algoritmo "Doacao Sangue"
2 Var
3   Nome : caracter
4   respIdade, respPeso, respTatu, respAlcool : caracter
5 Inicio
6
7   escreval("Qual seu Nome?")
8   leia(Nome)
9
10  escreva("Você confirma ter entre 19 e 69 anos de idade?")
11  escreval("(Utilize S ou N)")
12  leia(respIdade)
13
14  escreva("Você confirma pesar ao menos 50 kg?")
15  escreval("(Utilize S ou N)")
16  leia(respPeso)
17
18  escreva("Você confirma não ter tatuagem?")
19  escreval("(Utilize S ou N)")
20  leia(respTatu)
21
22  escreva("Você confirma não ter ingerido bebida alcoolica?")
23  escreval("(Utilize S ou N)")
24  leia(respAlcool)
25
```

```
se ((respIdade="N") ou (respPeso="N") ou (respTatu="N") ou (respAlcool="N")) entao  
    escreva(NOME, " infelizmente você não pode Doar Sangue!")  
senao  
    escreva(NOME, " PARABÉNS, VOCÊ PODE DOAR SANGUE!")  
fimse
```

```
25  
26 se ((respIdade="N") ou (respPeso="N") ou (respTatu="N") ou (respAlcool="N")) entao  
27     escreva(NOME, " infelizmente você não pode Doar Sangue!")  
28 senao  
29     escreva(NOME, " PARABÉNS, VOCÊ PODE DOAR SANGUE!")  
30 fimse  
31  
32  
33 Fimalgoritmo  
34
```



## EXERCÍCIO 3 - MAIS ESTRUTURADO

Área dos algoritmos ( Edição do código fonte ) -> Nome do arquivo: [DOACAO SANGUE.ALG]

```
1 Algoritmo "Doacao Sangue"  
2 Var  
3   Nome : caracter  
4   resp : caracter  
5 Inicio  
6  
7   escreval("Qual seu Nome?")  
8   leia(Nome)  
9  
10  escreva("Você tem entre 19 e 69 anos de idade?")  
11  escreval("(Utilize S ou N)")  
12  leia(resp)  
13
```

```

14  se(resp="S") entao
15      escreva("Você pesar ao menos 50 kg?")
16      escreval("(Utilize S ou N)")
17      leia(resp)
18
19      se(resp="S") entao
20          escreva("Você fez tatuagem?")
21          escreval("(Utilize S ou N)")
22          leia(resp)
23
24          se(resp="N") entao
25
26              escreva("Você ingeriu alcool?")
27              escreval("(Utilize S ou N)")
28              leia(resp)
29
30              se(resp="S") entao
31                  escreva("Infelizmente você não pode Doar Sangue!")
32
33              senao
34                  escreva("PARABÊNS, VOCÊ PODE DOAR SANGUE!")
35              fimse
36
37          senao
38              escreva("Infelizmente você não pode Doar Sangue!")
39          fimse
40
41      senao
42          escreva("Infelizmente você não pode Doar Sangue!")
43      fimse
44
45  senao
46      escreva("Infelizmente você não pode Doar Sangue!")
47  fimse

```

4 - Crie um software que ajude a polícia rodoviária federal calcular o valor das multas dos motoristas. O software deve perguntar ao usuário qual a velocidade permitida e qual a velocidade do motorista.

Se o motorista estiver acima da velocidade permitida, o software deve informar qual o tipo de infração e qual o valor da multa. Se o motorista estiver dentro do limite permitido, o programa encerra dizendo que ele não precisa pagar multa. Veja abaixo os valores de multas reajustados de infração por excesso de velocidade:

- Exceder velocidade em até 20% (infração média) = R\$85,00 + 4 pontos ;
- Exceder velocidade de 20 até 50% (infração grave) = R\$127,00 + 5 pontos;
- Exceder velocidade acima de 50% (infração gravíssima) = R\$574,00 + 7 pontos + apreensão da carteira + suspensão do direito de dirigir.

```
1 Algoritmo "Calculo de Multas"
2 // Disciplina : [Linguagem e Lógica de Programação]
3 // Professor : Adjenor Cristiano
4 // Autor(a) : Rafael Henrique Elias
5 // Data atual : 03/03/2023
6 Var
7
8 velocidade_permitida: real
9 velocidade_motorista: real
10 excesso_velocidade: real
11 tipo_multa: caractere
12 valor_multa: real
13 pontos_multa: real
14
15 Inicio
16
17 escreval("Bem vindo Policial. Para calcular o valor de multas, informe abaixo a velocidade
18 escreval
19 leia(velocidade_permitida)
20 escreval
21 escreval("Agora digite a velocidade que o motorista estava na via. (em km/h)")
22 escreval
23 leia(velocidade_motorista)
24 escreval
```

```

20 escreval
21 escreval("Agora digite a velocidade que o motorista estava na via. (em km/h)")
22 escreval
23 leia(velocidade_motorista)
24 escreval
25 se (velocidade_motorista > velocidade_permitida) Entao
26     excesso_velocidade <- ((velocidade_motorista / velocidade_permitida -1) * 100)
27     se excesso_velocidade > 0 entao
28         se excesso_velocidade <= 20 entao
29             tipo_multa <- "Média"
30             valor_multa <- 85
31             pontos_multa <- 4
32         senao
33             se (excesso_velocidade <= 50)entao
34                 tipo_multa <- "Grave"
35                 valor_multa <- 127
36                 pontos_multa <- 5
37             senao se (excesso_velocidade >= 51) entao
38                 tipo_multa <- "Gravissima"
39                 valor_multa <- 574
40                 pontos_multa <- 7
41             escreval
42             Escreval("A carteira de motorista será apreendida e o direito de dirigir
43         fimSe
44     fimSe
45 fimse
46 Escreval("Você excedeu a velocidade permitida em " ,excesso_velocidade, "%.")
47 Escreval("Tipo de infração: " , tipo_multa)
48 Escreval("Valor da multa: R$" , valor_multa , ",00")
49 Escreval("Pontos na carteira: " , pontos_multa)
50 Senão
51     Escreval("Você está dentro do limite de velocidade permitido na via. Não há multa a
52 FimSe
53 Fimalgoritmo

```



5 - Uma loja de salgadinhos precisa de ajuda para calcular o valor da conta dos clientes. Crie um programa que leia o código do produto que o cliente comprou e a quantidade e informe a ele quanto ele deverá pagar. Faça uso da estrutura ESCOLHA-CASO. A Tabela abaixo apresenta o código, a descrição e o preço de cada salgadinho.

Código	Descrição	Preço unitário
1	Coxinha	0,50
2	Pastel	0,75
3	Pão de queijo	0,40
4	Enroladinho	0,80

```
1  Algoritmo "Lanchonete"
2
3  Var
4  totalConta : REAL
5  opcao: INTEIRO
6  produtos : CARACTER
7  qtd: INTEIRO
8
9  Inicio
10 totalConta <- 0
11 produtos<- ""
12 ESCREVAL("*****")
13 ESCREVAL("                PASTELARIA PROGRAMANDO                ")
14 ESCREVAL("*****")
15 ESCREVAL("")
16 ESCREVAL("-----")
17 ESCREVAL("                CARDÁPIO:                ")
18 ESCREVAL("")
19 ESCREVAL("1 - COXINHA:                R$ 6,00")
20 ESCREVAL("2 - PASTEL:                R$ 5,00")
21 ESCREVAL("3 - EMPADA:                R$ 5,50")
22 ESCREVAL("4 - PÃO DE QUEIJO: R$ 4,50")
23 ESCREVAL("-----")
24 ESCREVAL("")
25 ESCREVAL("O QUE GOSTARIA DE PEDIR?")
26 LEIA(opcao)
```

```
27
28 ESCOLHA(opcao)
29     CASO 1
30         ESCREVAL("ÓTIMA ESCOLHA! QUANTAS COXINHAS VOCÊ DESEJA?")
31         LEIA(qtd)
32         ESCREVAL("")
33         ESCREVAL("RESUMO DO PEDIDO:")
34         ESCREVAL(qtd, " - COXINHA(S):          R$ ", (qtd*6):6:2)
35     CASO 2
36         ESCREVAL("ÓTIMA ESCOLHA! QUANTAS PASTEIS VOCÊ DESEJA?")
37         LEIA(qtd)
38         ESCREVAL("")
39         ESCREVAL("RESUMO DO PEDIDO:")
40         ESCREVAL(qtd, " - PASTEL(IS):          R$ ", (qtd*5):6:2)
41     CASO 3
42         ESCREVAL("ÓTIMA ESCOLHA! QUANTAS EMPADAS VOCÊ DESEJA?")
43         LEIA(qtd)
44         ESCREVAL("")
45         ESCREVAL("RESUMO DO PEDIDO:")
46         ESCREVAL(qtd, " - EMPADAS(S):          R$ ", (qtd*5.5):6:2)
47     CASO 4
48         ESCREVAL("ÓTIMA ESCOLHA! QUANTOS PÃES DE QUEIJO VOCÊ DESEJA?")
49         LEIA(qtd)
50         ESCREVAL("")
51         ESCREVAL("RESUMO DO PEDIDO:")
52         ESCREVAL(qtd, " - PÃO(ES) DE QUEIJO:      R$ ", (qtd*4.5):6:2)
53     FIMESCOLHA
54 Fimalgoritmo
```



# Escolha-Caso





## Escolha-Caso

Com as estruturas condicionais podemos verificar várias condições para várias possibilidades de saídas.

Porém, quando existem muitas condições, ficar usando várias estruturas se, uma dentro da outra, pode ser muito trabalhoso, além de dificultar a leitura e manutenção do código, por parte de outros programadores.



## Escolha-Caso

Para evitar isso, temos a opção de usar a estrutura ESCOLHA-CASO.

O ESCOLHA-CASO agiliza comparações quando existem muitas comparações com valores pré-estabelecidos. Basicamente, se verifica se uma variável está com o mesmo valor que algum dos casos definidos no código. Quando isso acontecer, executa-se o código referente a esse caso.



## Escolha-Caso

Imagine se precisamos escrever um código que recebe um número e imprima na tela seu valor por extenso.

Podemos utilizar a estrutura condicional SE e nosso código ficaria assim:

```
1 algoritmo "Número por Extenso"
2 var
3   numero : INTEIRO
4 inicio
5
6   ESCREVA ("Digite um número entre 1 e 3: ")
7   LEIA (numero)
8
9   SE (numero = 1) ENTÃO
10    ESCREVA (numero, " = UM")
11  SENÃO
12    SE (numero = 2) ENTÃO
13      ESCREVA (numero, " = DOIS")
14    SENÃO
15      SE (numero = 3) ENTÃO
16        ESCREVA (numero, " = TRÊS")
17      SENÃO
18        ESCREVA (numero, " = NÃO CADASTRADO")
19      FIMSE
20    FIMSE
21  FIMSE
22 FIMSE
23 |
24 fimalgoritmo
```



## Escolha-Caso

Para uma sequência de 3 valores ficaria até entendível, mas se aumentarmos os valores nosso código ficaria bem extenso.

Para facilitar essa situação temos a estrutura ESCOLHA-CASO ou SWITCH-CASE em outras Linguagens.


# Nosso código ficaria assim:

```
- Área dos programas ( Edição do código fonte ) -> Nome do arquivo: [NÚMERO POR EXTENSO ESCOLHA]

1 algoritmo "Número por Extenso ESCOLHA-CASO"
2 var
3     numero : INTEIRO
4 inicio
5
6     ESCREVA ("Digite um número entre 1 e 3: ")
7     LEIA (numero)
8
9     ESCOLHA (numero)
10        CASO 1
11            ESCREVA (numero, " = UM")
12        CASO 2
13            ESCREVA (numero, " = DOIS")
14        CASO 3
15            ESCREVA (numero, " = TRÊS")
16        OUTROCASO
17            ESCREVA (numero, " = NÃO CADASTRADO")
18    FIMESCOLHA
19
20 fimalgoritmo
```



Nosso RESULTADO ficaria assim:

 Console simulando o modo texto do MS-DOS

Digite um número entre 1 e 3: 3

3 = TRÊS

>>> Fim da execução do programa !






## Vamos praticar:

Desenvolva um software que receba do usuário a sigla correspondente a um estado de imprima da tela o nome do estado completo:

EX:

MG - Minas Gerais

SP - São Paulo




```
1 algoritmo "Estado - UF"
2 var
3     UF : Caractere
4 inicio
5
6     ESCREVA ("Digite o UF de um Estado: ")
7     LEIA (UF)
8     UF<-MAIUSC(UF)
9     ESCOLHA (UF)
10        CASO "MG"
11            ESCREVA (UF, " = Minas Gerais")
12        CASO "SP"
13            ESCREVA (UF, " = São Paulo")
14        CASO "RJ"
15            ESCREVA (UF, " = Rio de Janeiro")
16        CASO "ES "
17            ESCREVA (UF, " = Espírito Santo")
18        OUTROCASO
19            ESCREVA ("ESTADO NÃO CADASTRADO")
20    FIMESCOLHA
21
22 fimalgoritmo
```



Vamos praticar:

Desenvolva um software que simule uma calculadora.

Seu software deve receber dois valores e um sinal, depois realizar o cálculo e imprimir na tela o resultado.



## Calculadora - Explicação:

Para entender melhor, vamos construir passo-a-passo um algoritmo que vai simular uma calculadora com as operações básicas: adição, subtração, multiplicação e divisão.

1. Primeiro vamos declarar as variáveis, serão quatro: uma para cada número da operação do tipo real, uma para receber o tipo de operador (+, -, \* ou /) do tipo caractere e uma para receber o resultado da operação do tipo real.

# Calculadora

Área dos algoritmos ( Edição do código fonte ) -> Nome do arquivo: [semnome]


```
1 Algoritmo "Nome_Programa"  
2  
3 Var  
4     numero1, numero2: real  
5     operador: caractere  
6     resultado: real
```



Calculadora:


2. Depois, vamos pedir ao usuário os valores dos números, e o tipo de operação que vai ser:

```
7 Inicio
8
9     escreva("Digite o primeiro número")
10    leia(numero1)
11    escreva("Digite o tipo de operador [ +, -, *, / ]")
12    leia(operador)
13    escreva("Digite o segundo número")
14    leia(numero2)
15
```



3. Agora o mais importante, vamos usar a estrutura ESCOLHA-CASO para verificar o valor do operado recebido pelo usuário, em cada caso


```
15
16     escolha (operador)
17         caso "+"
18             resultado <- numero1 + numero2
19         caso "-"
20             resultado <- numero1 - numero2
21         caso "*"
22             resultado <- numero1 * numero2
23         caso "/"
24             resultado <- numero1 / numero2
25     fimescolha
26
```



4. Por último, escrevemos na tela o valor da variável resultado que foi calculada em algum dos casos do ESCOLHA.

```
26  
27     escreva ("O resultado é: ", resultado)  
28  
29 Fimalgoritmo
```





5. **MELHORANDO NOSSO CÓDIGO:** De acordo com a matemática, não é possível dividir nenhum número por Zero, então devemos proteger nosso sistema pra que evitar esse cálculo impossível:

```
23 caso "/"
24     se (numero2<>0) entao
25         resultado<- numero1/numero2
26     senao
27         escreval("É impossivel divisão por ZERO!")
28         Fimalgoritmo
29     fimse
```

## Portugol Studio

```
programa {
```

```
    funcao inicio() {
```

```
        real numero1, numero2
```

```
        caracter operador
```

```
        real resultado
```

```
        escreva ("Digite o primeiro número: ")
```

```
        leia(numero1)
```

```
        escreva ("Digite a operação: ")
```

```
        leia(operador)
```

```
        escreva ("Digite o segundo número: ")
```

```
        leia(numero2)
```

```
        escolha (operador) {
```

```
            caso '+':
```

```
                resultado = numero1 + numero2
```

```
            pare
```

```
            caso '-':
```

```
                resultado = numero1 - numero2
```

```
            pare
```

```
            caso '*':
```

```
                resultado = numero1 * numero2
```

```
            pare
```

```
            caso '/':
```

```
                resultado = numero1 / numero2
```

```
        }
```

```
        escreva("O resultado é: " + resultado)
```


```
    }
```

```
}
```



## Exercícios:

1. Crie um software que receba do usuário um valor inteiro e imprima na tela o mês do ano correspondente ao valor recebido (ex. recebendo o valor 1, o programa imprime “Janeiro”; recebendo o valor 2, o programa imprime “Fevereiro”...).
2. Crie um software que receba do usuário um valor inteiro de 1 a 10 e imprima na tela o número por extenso. Se o usuário digitar um número maior que 10 ele deve imprimir “Valor Inválido”.



3. Desenvolva um software que receba o nome do usuário, o sexo, a cidade e o ano que ele nasceu e depois imprima na tela sua ficha cadastral utilizando “Sr” para homens e “Sr<sup>a</sup>” para mulheres.

Considere apenas o ano para cálculo da idade (quem nasceu em 2002 considere que ele tem 22 anos, quem nasceu em 2001 considere 23...)


Ex:

Ficha Cadastral

Sr: Adjenor Cristiano Queiroz

Idade: 22 anos

Cidade: Pará de Minas



4 - (DESAFIO) Desenvolva um Software para um restaurante que irá mostrar para o usuário um cardápio de refeições (com no mínimo 4 opções) para escolha:

Ex:

1 - Risotto \_\_\_\_\_ R\$ 25,00

2 - Espaguete \_\_\_\_\_ R\$ 22,00

Depois pergunte a ele se ele deseja alguma bebida. Se ele desejar, mostre a ele um menu com as opções de bebida (ao menos 4).

Ex:

1 - Coca Cola \_\_\_\_\_ R\$ 5,00

2 - Água Mineral \_\_\_\_\_ R\$ 3,00

Ao final mostre a ele qual o seu pedido e o valor da conta.

Ex: PEDIDO:

- Espaguete R\$ 22,00 | - Coca Cola R\$5,00

TOTAL: R\$27,00.

Dica: Use uma variável do tipo caractere para armazenar os itens do pedido e uma variável real para armazenar os valores